
Subject: Re: Difficulty with N_ELEMENTS command
Posted by [cgguido](#) on Fri, 10 Feb 2012 04:21:03 GMT
[View Forum Message](#) <> [Reply to Message](#)

Try:

```
FOR m=0,nyears-1 DO BEGIN
    dummy=WHERE(precip_day[m,*] gt smoothed, n)
    exceed_freq[m]=n
ENDFOR
```

Hope it helps.

G

Subject: Re: Difficulty with N_ELEMENTS command
Posted by [cgguido](#) on Fri, 10 Feb 2012 04:28:06 GMT
[View Forum Message](#) <> [Reply to Message](#)

You could also try "the IDL" way:

```
s=size(precip_day, /dim)
smoothed2=rebin(smoothed, s, /sample)

test=precip_day gt smoothed2
exceed_freq=total(test, 1)
```

Look mum! No loops!!

G

Subject: Re: Difficulty with N_ELEMENTS command
Posted by [Andrew King](#) on Fri, 10 Feb 2012 04:32:52 GMT
[View Forum Message](#) <> [Reply to Message](#)

On Feb 10, 3:28 pm, Gianguido Cianci <gianguido.cia...@gmail.com>
wrote:

> You could also try "the IDL" way:
>
> s=size(precip_day, /dim)
> smoothed2=rebin(smoothed, s, /sample)
>

```
> test=precip_day gt smoothed2  
> exceed_freq=total(test, 1)  
>  
> Look mum! No loops!!  
>  
> G
```

Works perfectly now. Thanks very much.

Andrew

Subject: Re: Difficulty with N_ELEMENTS command
Posted by [Yngvar Larsen](#) on Sun, 12 Feb 2012 19:14:33 GMT
[View Forum Message](#) <> [Reply to Message](#)

On Feb 10, 5:03 am, Andrew King <andrewkin...@gmail.com> wrote:

```
> Hi.  
>  
> I'm trying to create an array of the number of elements that satisfy a  
> certain criterion as below.  
>  
> FOR m=0,nyears-1 DO BEGIN  
>   exceed_freq(m)=N_ELEMENTS(WHERE(precip_day(m,*) gt smoothed(*)))  
> ENDFOR  
>  
> The problem I have is that for some years in my program there are no  
> elements that satisfy this criterion, however the value returned in  
> the array 'exceed_freq' is 1 instead of 0.  
>  
> I would appreciate any comments.
```

A one-liner solution (mostly because it fun :) could be

```
exceed_freq = total(precip_day gt  
transpose(smoothed[*,lindgen(nyears)]), 2)
```

or

```
exceed_freq = total(transpose(precip_day) gt  
smoothed[*,lindgen(nyears)], 1)
```

--
Yngvar

Subject: Re: Difficulty with N_ELEMENTS command

Posted by [Yngvar Larsen](#) on Sun, 12 Feb 2012 19:21:30 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Feb 12, 8:14 pm, Yngvar Larsen <larsen.yng...@gmail.com> wrote:

```
> A one-liner soulution (mostly because it fun :) could be
>
> exceed_freq = total(precip_day gt
> transpose(smoothed[*,lindgen(nyears)]), 2)
>
> or
>
> exceed_freq = total(transpose(precip_day) gt
> smoothed[*,lindgen(nyears)], 1)
>
> --
> Yngvar
```

Ironically, Google Groups decided that my one-liners really should be two-liners :)

--
Yngvar
